```
Needs["ErrorBarPlots`"]

SetDirectory[NotebookDirectory[]]
```

```
/Users/miham/Desktop/ISR/RadiusFit
```

Data

```
M = 0.93827;
mup = 2.79;
ħc = 0.197327;
```

$$\text{Tau}[Q2\_] := \frac{Q2}{4\,M^2}$$

$$\text{Eps}[th\_, Q2\_] := \left(1 + 2\,(1 + \text{Tau}[Q2])\,\text{Tan}\left[\frac{th}{2}\right]^2\right)^{-1}$$

$$\text{GD}[x\_, a\_, n\_] := \frac{1}{\left(1 + \frac{x}{a}\right)^2} * n$$

$$\text{GM}[Q2\_, R\_, n\_, a\_, b\_] := n\left(1 - \frac{1}{6 * ħc^2} R^2\,Q2 + \frac{a}{120 * ħc^4} Q2^2 - \frac{b}{5040 * ħc^6} Q2^3\right)$$

```
data2 = Select[Map[{#[[1]], #[[2]], #[[3]]} &,
    Import["ISRFormFactorNoEl.dat"]], #[[1]] < 0.03 &];
```

```
data2;
```

```
data = Join[Select[data2, (#[[1]] > 0.0 && #[[1]] < 0.1) &]];
```

```
s1 = Show[ErrorListPlot[Map[{{#[[1]], #[[2]]}, ErrorBar[#[[3]]]} &, data],
   PlotStyle → Hue[0]], ErrorListPlot[
   Map[{{#[[1]], #[[2]]}, ErrorBar[#[[3]]]} &, data2], PlotStyle → Hue[0.6]]]
```



Manual Fit with Chi2

```
Chi2b[R_, n_, a_, b_, data_] :=
 1/(Length[data] - 2) Sum[(GM[data[[i, 1]], R, 1, a, b] - n * data[[i, 2]])^2 /
    (n^2 * data[[i, 3]]^2), {i, 1, Length[data]}]

min = FindMinimum[Chi2b[RR, nnn, 2.59, 29.8, data], {RR, nnn}]
```

FindMinimum:lstol:
The line search decreased the step size to within the tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the function. You may need more than MachinePrecision digits of working precision to meet these tolerances. >>

```
{2.99202, {RR → 0.855357, nnn → 1.00143}}
```

```
Clear[r]
```

```
α2[data_, RR_, nnn_, a_, b_] :=
 Sum[1/data[[i, 3]]^2 {{D[GM[Q2, r, 1/c, a, b], r] D[GM[Q2, r, 1/c, a, b], r],
     D[GM[Q2, r, 1/c, a, b], r] D[GM[Q2, r, 1/c, a, b], c]},
    {D[GM[Q2, r, 1/c, a, b], c] D[GM[Q2, r, 1/c, a, b], r],
     D[GM[Q2, r, 1/c, a, b], c] D[GM[Q2, r, 1/c, a, b], c]}} /.
   {r → RR, c → nnn, Q2 → data[[i, 1]]}, {i, 1, Length[data]}]
```

```
RadiusGM2[{a_, n_}, cc_] := Module[{},
  dd = D[GM[x, R, 1, 2.59, 29.8], x] /. {x → 0};
  ra = Sqrt[-6 * 0.197326^2 * dd];
  dr = Sqrt[(D[ra, R])^2 cc[[1, 1]]];
  {ra, dr, n1, Sqrt[cc[[2, 2]]]} /. {R → a, n1 → n}
 ]
```

```
FindMyRadius[a_, b_, data_] := Module[{},
  min = FindMinimum[Chi2b[RR, nnn, a, b, data], {RR, nnn}];
  (*Print[min];*)
  Cov = Inverse[α2[data, RR, nnn, a, b] /. min[[2]]];
  (*Print[Cov];*)
  RadiusGM2[{RR, nnn} /. min[[2]], Cov]
 ]
```

```
res = FindMyRadius[2.59, 29.8, data]
```

FindMinimum:lstol:
The line search decreased the step size to within the tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the function. You may need more than MachinePrecision digits of working precision to meet these tolerances. >>

```
{0.855353, 0.0191881, 1.00143, 0.00119756}
```

Radius Model dependent Error band

```
ModelDepRadiusError[a_, da_, b_, db_, NN_] := Module[{},
  alist = RandomVariate[NormalDistribution[a, da], NN];
  blist = RandomVariate[NormalDistribution[b, db], NN];
  fulllist = Table[{alist[[i]], blist[[i]]}, {i, 1, NN}];
  ll = Map[FindMyRadius[#[[1]], #[[2]], data] &, fulllist];
  meanr = Mean[Map[#[[1]] &, ll]];
  sigmar = Sqrt[Variance[Map[#[[1]] &, ll]]];
  meann = Mean[Map[#[[3]] &, ll]];
  sigman = Sqrt[Variance[Map[#[[3]] &, ll]]];
  {meanr, sigmar, meann, sigman}
 ]
```

```
ModelDepRadiusError[2.59, 0.194, 29.8, 14.71, 100]
```

FindMinimum:lstol:
The line search decreased the step size to within the tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the function. You may need more than MachinePrecision digits of working precision to meet these tolerances ≫

FindMinimum:lstol:
The line search decreased the step size to within the tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the function. You may need more than MachinePrecision digits of working precision to meet these tolerances ≫

FindMinimum:lstol:
The line search decreased the step size to within the tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the function. You may need more than MachinePrecision digits of working precision to meet these tolerances ≫

General::stop: Further output of FindMinimum:lstol will be suppressed during this calculation ≫

```
{0.854989, 0.00248736, 1.00144, 0.0000585938}
```

Manual Form - Factor Error band

```
FormFactor[Q2_, NN_, R_, dR_, a_, da_, b_, db_] := Module[{},
  rList = RandomVariate[NormalDistribution[R, dR], NN];
  aList = RandomVariate[NormalDistribution[a, da], NN];
  bList = RandomVariate[NormalDistribution[b, db], NN];
  fulllist = Table[{rList[[i]], aList[[i]], bList[[i]]}, {i, 1, NN}];
  FF = Map[GM[Q2, #[[1]], 1, #[[2]], #[[3]]] &, fulllist];
  mean = Mean[FF];
  sigma = Sqrt[Variance[FF]];
  {Q2, mean, sigma}
 ]
```

```
FormFactor[0.01, 100, res[[1]], res[[2]], 2.59, 0.194, 29.8, 14.71]
```

```
{0.01, 0.969967, 0.00139184}
```

```
FFTable = Table[FormFactor[Q2, 100 000, res[[1]],
    res[[2]], 2.59, 0.194, 29.8, 14.71], {Q2, 0.001, 0.2, 0.001}];
```

```
ListPlot[Map[{#[[1]], #[[2]]} &, FFTable]]
```




```
Export["ResultsISRStandAloneManualNew.dat", FFTable];
```

Analytical Calculation of Model error!

```
Clear[GE, RR, R, a, b, n, Q2, dRR]
```

$$\text{RR} = \text{Sqrt}\left[\left(\frac{-\text{GE}}{\text{n}} + 1 + \frac{\text{a}}{120 * \hbar c^4}\,\text{Q2}^2 - \frac{\text{b}}{5040 * \hbar c^6}\,\text{Q2}^3\right)\frac{6\,\hbar c^2}{\text{Q2}}\right]$$

$$0.48335\sqrt{\frac{1 - \frac{\text{GE}}{\text{n}} + 5.49633\,\text{a}\,\text{Q2}^2 - 3.36086\,\text{b}\,\text{Q2}^3}{\text{Q2}}}$$

```
dRR[aa_, da_, bb_, db_, GGE_, nn_, QQ2_] :=
 Sqrt[(D[RR, a] da)^2 + (D[RR, b] db)^2] /. {a → aa, b → bb, GE → GGE, n → nn, Q2 → QQ2}

dRR[2.59, 0.194, 29.8, 14.71, 0.99, 1, 0.017]
```

```
0.00623141
```

```
qq = Map[{#[[1]], dRR[2.59, 0.194, 29.8, 14.71,
      GM[#[[1]], 0.85535281, 1.001439123, 2.59, 29.8], 1.0014, #[[1]]]} &, data]
```

```
{{0.00257314, 0.000378271}, {0.00210662, 0.000309138},
 {0.00180739, 0.000265027}, {0.001556, 0.000228087}, {0.00133657, 0.000195918},
 {0.0069408, 0.00106271}, {0.00596758, 0.00090258}, {0.00510955, 0.000765572},
 {0.00440878, 0.000656202}, {0.0037858, 0.000560633}, {0.00327766, 0.000483687},
 {0.00283015, 0.000416573}, {0.00244682, 0.000359504}, {0.00244627, 0.000359422},
 {0.00210893, 0.000309479}, {0.00182937, 0.000268262}, {0.0153986, 0.00275627},
 {0.0128783, 0.00218524}, {0.0110455, 0.0018081}, {0.0095245, 0.00151715},
 {0.00819877, 0.00127821}, {0.00708697, 0.00108723}, {0.0061315, 0.000929181}}
```

```
ListPlot[qq]
```



The model uncertainty is detemined by the uncertainty of the point at the largest Q2, because there the effect is the largest!

Systematic Uncertainty

495 MeV

```
Syst495MeVData = Import["ISRFFSystematicalError495MeV.dat"]
s495 = Interpolation[Map[{#[[1]], #[[2]] / 100} &, Syst495MeVData]];
```

```
{{0.0170417, 0.445293, 0}, {0.0153986, 0.444452, 0},
 {0.0128783, 0.418822, 0}, {0.0110455, 0.393569, 0}, {0.0095245, 0.37142, 0},
 {0.00819877, 0.389644, 0}, {0.00708697, 0.559909, 0}, {0.0061315, 0.670987, 0}}
```

```
s495[0.01]
```

```
0.00375348
```

```
Data495MeV = Select[Map[{#[[1]], #[[2]], #[[3]]} &,
    Import["ISRFormFactor495MeV.dat"]], #[[1]] < 0.017 &]
```

```
{{0.0153986, 0.956278, 0.0025899},
 {0.0128783, 0.960465, 0.00172386}, {0.0110455, 0.96911, 0.00188524},
 {0.0095245, 0.966328, 0.0016995}, {0.00819877, 0.97523, 0.00245552},
 {0.00708697, 0.975, 0.00204589}, {0.0061315, 0.978237, 0.00203728}}
```

```
Combined495MeV = Map[{#[[1]], #[[2]], #[[3]], s495[#[[1]]]} &, Data495MeV]
```

```
{{0.0153986, 0.956278, 0.0025899, 0.00444452},
 {0.0128783, 0.960465, 0.00172386, 0.00418822},
 {0.0110455, 0.96911, 0.00188524, 0.00393569},
 {0.0095245, 0.966328, 0.0016995, 0.0037142},
 {0.00819877, 0.97523, 0.00245552, 0.00389644},
 {0.00708697, 0.975, 0.00204589, 0.00559909},
 {0.0061315, 0.978237, 0.00203728, 0.00670987}}
```

330 MeV

```
Syst330MeVData = Import["ISRFFSystematicalError330MeV.dat"]
s330 = Interpolation[Map[{#[[1]], #[[2]] / 100} &, Syst330MeVData]];
```

```
{{0.0078145, 1.09457, 0}, {0.0069408, 1.08763, 0}, {0.00596758, 1.02248, 0},
 {0.00510955, 0.96166, 0}, {0.00440878, 0.900458, 0}, {0.0037858, 0.877308, 0},
 {0.00327766, 0.85083, 0}, {0.00283015, 0.790093, 0}, {0.00244682, 0.803002, 0},
 {0.00244627, 0.802977, 0}, {0.00210893, 0.868517, 0}, {0.00182937, 0.878239, 0}}
```

```
s330[0.005]
```

```
0.00951384
```

```
Data330MeV = Select[Map[{#[[1]], #[[2]], #[[3]]} &,
    Import["ISRFormFactor330MeV.dat"]], #[[1]] < 0.007 &]
```

```
{{0.0069408, 0.982376, 0.00410094},
 {0.00596758, 0.972034, 0.00305643}, {0.00510955, 0.982733, 0.00297362},
 {0.00440878, 0.979775, 0.00336679}, {0.0037858, 0.988231, 0.00312875},
 {0.00327766, 0.988665, 0.00354716}, {0.00283015, 0.99531, 0.00393255},
 {0.00244682, 0.997295, 0.0040697}, {0.00244627, 0.99814, 0.00459216},
 {0.00210893, 0.999881, 0.00404735}, {0.00182937, 0.98009, 0.00455452}}
```

```
Combined330MeV = Map[{#[[1]], #[[2]], #[[3]], s330[#[[1]]]} &, Data330MeV]
```

```
{{0.0069408, 0.982376, 0.00410094, 0.0108763},
 {0.00596758, 0.972034, 0.00305643, 0.0102248},
 {0.00510955, 0.982733, 0.00297362, 0.0096166},
 {0.00440878, 0.979775, 0.00336679, 0.00900458},
 {0.0037858, 0.988231, 0.00312875, 0.00877308},
 {0.00327766, 0.988665, 0.00354716, 0.0085083},
 {0.00283015, 0.99531, 0.00393255, 0.00790093},
 {0.00244682, 0.997295, 0.0040697, 0.00803002},
 {0.00244627, 0.99814, 0.00459216, 0.00802977},
 {0.00210893, 0.999881, 0.00404735, 0.00868517},
 {0.00182937, 0.98009, 0.00455452, 0.00878239}}
```

195 MeV

```
Syst195MeVData = Import["ISRFFSystematicalError195MeV.dat"]
s195 = Interpolation[Map[{#[[1]], #[[2]] / 100} &, Syst195MeVData]];
```

```
{{0.00273397, 1.02731, 0}, {0.00257314, 1.00992, 0}, {0.00210662, 0.991071, 0},
 {0.00180739, 0.966454, 0}, {0.001556, 1.02105, 0}, {0.00133657, 1.01279, 0}}
```

```
s195[0.002]
```

```
0.00976497
```

```
Data195MeV = Select[Map[{#[[1]], #[[2]], #[[3]]} &,
    Import["ISRFormFactor195MeV.dat"]], #[[1]] < 0.0027 &]
```

```
{{0.00257314, 1.00014, 0.00471763},
 {0.00210662, 0.994993, 0.00334811}, {0.00180739, 0.982284, 0.00332201},
 {0.001556, 0.99599, 0.00336262}, {0.00133657, 1.00407, 0.0039531}}
```

```
Combined195MeV = Map[{#[[1]], #[[2]], #[[3]], s195[#[[1]]]} &, Data195MeV]
```

```
{{0.00257314, 1.00014, 0.00471763, 0.0100992},
 {0.00210662, 0.994993, 0.00334811, 0.00991071},
 {0.00180739, 0.982284, 0.00332201, 0.00966454},
 {0.001556, 0.99599, 0.00336262, 0.0102105},
 {0.00133657, 1.00407, 0.0039531, 0.0101279}}
```

```
CombinedAll = Join[Combined495MeV, Combined330MeV, Combined195MeV];
```

Analysis

```
RandomizeData[data_] := Module[{},
  NewData =
   Map[{#[[1]], RandomVariate[NormalDistribution[#[[2]], #[[4]]], 1][[1]],
       #[[3]]} &, data];
  NewData
 ]
```

```
res = FindMyRadius[2.59, 29.8, data]
```

FindMinimum:lstol:
The line search decreased the step size to within the tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the function. You may need more than MachinePrecision digits of working precision to meet these tolerances ≫

```
{0.855353, 0.0191881, 1.00143, 0.00119756}
```

```
Length[data]
```

```
23
```

```
Length[RandomizeData[CombinedAll]]
```

```
23
```

```
FindMyRadius[2.59, 29.8, RandomizeData[CombinedAll]]
```

FindMinimum:lstol:
The line search decreased the step size to within the tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the function. You may need more than MachinePrecision digits of working precision to meet these tolerances ≫

```
{0.812525, 0.0202958, 1.00394, 0.00120357}
```

```
Show[ErrorListPlot[Map[{{#[[1]], #[[2]]}, ErrorBar[#[[3]]]} &, CombinedAll],
  PlotStyle → Hue[0]], ErrorListPlot[Map[{{#[[1]], #[[2]]}, ErrorBar[#[[3]]]} &,
   RandomizeData[CombinedAll]], PlotStyle → Hue[0.6]]]
```



```
SysPoints = Table[
   Join[{i}, FindMyRadius[2.59, 29.8, RandomizeData[CombinedAll]]], {i, 1, 500}];
```

FindMinimum:lstol:
The line search decreased the step size to within the tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the function. You may need more than MachinePrecision digits of working precision to meet these tolerances ≫

FindMinimum:lstol:
The line search decreased the step size to within the tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the function. You may need more than MachinePrecision digits of working precision to meet these tolerances ≫

FindMinimum:lstol:
The line search decreased the step size to within the tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the function. You may need more than MachinePrecision digits of working precision to meet these tolerances ≫

General::stop: Further output of FindMinimum:lstol will be suppressed during this calculation ≫

```
ErrorListPlot[
 Map[{{#[[1]], #[[2]]}, ErrorBar[#[[3]]]} &, SysPoints], PlotStyle → Hue[0]]
```




```
mean = Mean[Map[#[[2]] &, SysPoints]]
sigma = Sqrt[Variance[Map[#[[2]] &, SysPoints]]]
```

```
0.852881
```

```
0.042325
```

```
Show[
 ErrorListPlot[Map[{{#[[1]] + 0.0001, #[[2]]}, ErrorBar[#[[3]]]} &, CombinedAll],
  PlotStyle → Hue[0.2]],
 ErrorListPlot[Map[{{#[[1]] - 0.0002, #[[2]]}, ErrorBar[#[[3]]]} &,
   RandomizeData[CombinedAll]], PlotStyle → Hue[0.6]],
 ErrorListPlot[Map[{{#[[1]], #[[2]]}, ErrorBar[#[[3]]]} &, data],
  PlotStyle → Hue[0]], ErrorListPlot[
  Map[{{#[[1]], #[[2]]}, ErrorBar[#[[3]]]} &, data2], PlotStyle → Hue[0.6]]]
```




```
Integrate[t/x (x/F)^t, {x, 0, y}]
```

```
ConditionalExpression[(y/F)^t, Re[t] > 0]
```

In[4]:=
```
(0.005/0.5)^0.02
```
Out[4]= 0.912011

In[5]:=
```
(0.005/0.3)^0.02
```
Out[5]= 0.921376

In[6]:=
```
(0.005/0.2)^0.02
```
Out[6]= 0.928878